

3 圧延理論－2 (Orowan の理論・板プロファイルの形成機構)

東京大学生産技術研究所 柳本 潤

1. はじめに

圧延理論－2 では、Orowan の圧延理論とその数値解析について詳しく述べる。ついで、板プロファイルの形成機構とその解析に利用されている理論（分割モデル、スリットモデル）についてもふれる。なお、方程式番号、参考文献番号は初級コースからの通番となっているので、是非初級コースのテキストもご一読頂きたい。

2. Orowan の圧延理論

Orowan の圧延理論⁴⁾は、初等解析法に基づく2次元圧延理論としてはKarmanの理論と並び最も基本的かつ代表的な理論である。Karmanの方程式¹⁾については、Nadaiによって解析解（正確に言えば近似解ではあるが）が与えられており²⁾、これについては初級コースにおいて詳しく説明した。Orowanの圧延理論の基礎方程式は⁴⁾、Simsの理論⁵⁾の様に、付着摩擦状態については解が得られているが、一般の場合について解析的な解を得ることができない。ただ、現在では数値解析を容易に実施することができる状況にあり、混合摩擦状態をも含め正確な解を得ることができる。以後の説明では、圧延方向を x 軸（原点は軸心直下）、板厚方向を y 軸（原点は板厚中央）とする。

2.1. 前提となる条件

Karmanの圧延理論¹⁾では、ロールバイト内での xy 面内せん断応力 τ_{xy} を無視している。従って圧延方向応力 σ_{xx} ・板厚方向応力 σ_{yy} の分布は板厚方向に均一であり、故に均一変形理論と呼ばれている。Orowanの圧延理論⁴⁾では、 xy 面内でのせん断応力 τ_{xy} （または σ_{xy} ）が、ロールとの接触面では摩擦応力 τ_f に等しく、内部ではロール表面からの距離に応じて減少し、板厚中央では0となると仮定する。従って、板厚方向応力 σ_{yy} （正確には円周方向応力 s ）が後述するスラブ要素内において均一であると仮定すると、降伏条件より圧延方向応力は板厚方向に分布を持つ。これらのことがKarmanの圧延理論との大きな相違であり、故にKarmanの圧延理論が「均一変形理論」と呼ばれるのに対し、Orowanの圧延理論は「不均一変形理論」もしくは「一般理論」と呼ばれている。

2.2. 圧延方向力の釣り合い条件

図9に示す、ロールバイト内微小要素についての、圧延方向力の釣り合い条件は、(27)式により表される。

$$\frac{df(\phi)}{d\phi} = 2R \cdot p(\phi) \sin \phi \mp 2R \cdot \tau_f \cos \phi = 2R \cdot p(\phi) (\sin \phi \mp \mu \cos \phi) \quad (27)$$

ただし、 ϕ は位置角、 $f(\phi)$ はスラブ要素に作用する圧延方向合力（圧縮側を正とする）、 $p(\phi)$ はロールよりスラブ要素に作用する鉛直方向応力（圧縮側を正とする）である。以後符号 \pm, \mp は、上側は後進域、下側は先進域を表すものとする。

なお、(27)式にて表される圧延方向力の釣り合い条件は、実は、圧延理論（初級）にて説明した Karman 理論と同じ意味を持つ。すなわち、 $\theta \Leftrightarrow \phi$ 、 $h\sigma_{xx} \Leftrightarrow -f(\phi)$ 、 $dx/\cos\theta \Leftrightarrow Rd\phi$ であるから、Karman 理論と Orowan 理論で利用する釣り合い式は同じであると考えて良い。




図9 圧延方向力の釣り合い条件

2.3. 水平方向合力と鉛直方向応力との関係

Orowan の圧延理論では、Karman の理論では無視されていたせん断応力 τ が存在するものとしている。せん断応力は板厚方向の一次分布を仮定しており、例えば図9のロールバイト内では、

$$\tau = \frac{y}{h/2} \tau_f = \frac{y}{h/2} \mu p \quad (27-2)$$

なる分布、すなわち、ロール表面でのせん断応力は摩擦応力に等しく、板厚中心ではゼロであるとの分布を仮定している。板厚方向、圧延方向の応力をそれぞれ σ_r 、 σ_l とすれば、平面ひずみ状態での降伏条件は、

$$(\sigma_l - \sigma_t)^2 + 4\tau^2 = 4k^2 \quad (27-3)$$

と書ける¹⁶⁾. 因みに Karman の理論では, せん断応力を無視した場合について, 降伏条件式 $\sigma_1 - \sigma_3 = \sigma_t - \sigma_l = 2k$ を適用している. 式(27-3)より, 圧延方向応力 σ_l を求めると,

$$\sigma_l = \sigma_t - 2k \sqrt{1 - \left(\frac{\tau}{k}\right)^2} = \sigma_t - 2k \sqrt{1 - \left(\frac{y}{h/2} \frac{\mu \sigma_t}{k}\right)^2} \quad (28)$$

あるいは, 図 10 に示されている角度 θ, ϕ を用いて表示すれば,

$$\sigma_l = \sigma_t - 2k \sqrt{1 - a^2 \left(\frac{\theta}{\phi}\right)^2}, \quad a \equiv \frac{\mu}{k} \sigma_t \quad (29-1)$$

$$\tau = -\frac{\mu \phi}{\phi} \theta \quad (29-2)$$

これは傾斜 2 平面による圧縮に関する Nadai の解¹とも呼ばれるが, 実態は降伏条件そのものである。




図10 半径方向応力と円周方向応力との関係

k はせん断降伏応力, μ は摩擦係数である. a は被圧延材とロールとのすべり状態を表

¹ この解は, 平行 2 平板による圧縮についての Prandtl の解を発展させたものである

すパラメータであり、すべり摩擦の場合には $a \leq 1$ の値をとり、 $a \geq 1$ は固着摩擦状態、すなわち摩擦応力=被圧延材のせん断降伏応力、である場合に対応する。

(29)式をもとに水平方向合力 $f(\phi)$ と鉛直方向応力 $p(\phi)$ との関係を求める手順は以下の通りである。水平方向合力 $f(\phi)$ は、円弧を含む面上に作用している応力成分 σ_t, τ より、 σ_t の水平方向分応力と接線方向応力 τ の水平方向分応力を円周全長について積分したものである。半径方向応力 σ_r の水平方向分応力を円周全長について積分することにより、半径方向応力 σ_r に起因する水平力 $f_r(\phi)$ は以下の式により得られる。

$$f_t(\phi) = 2 \int_0^\phi \sigma_t \cos \theta \frac{h}{2 \sin \phi} d\theta \quad (30)$$

せん断応力 τ に起因する水平力 $f_\tau(\phi)$ は以下の通りに与えられる。

$$f_\tau(\phi) = 2 \int_0^\phi \pm \tau \sin \theta \frac{h}{2 \sin \phi} d\theta \quad (32)$$

$f_t(\phi)$ の計算は以下の通りである。(30)式に(29-1)式を代入すると次式が得られる。

$$f_t(\phi) = hp - \frac{2hk}{\sin \phi} \int_0^\phi \sqrt{1 - a^2 \left(\frac{\theta}{\phi}\right)^2} \cos \theta d\theta = hp - 2hk\varpi(\phi, a) \quad (31)$$

$$\varpi(\phi, a) \equiv \frac{1}{\sin \phi} \int_0^\phi \sqrt{1 - a^2 \left(\frac{\theta}{\phi}\right)^2} \cos \theta d\theta$$

式(31)による $\varpi(\phi, a)$ の計算では、円周方向応力 σ_t は θ によらず一定であること、すなわち $\sigma_t = \sigma_r(\phi) = p(\phi)$ であることに注意されたい。

$f_\tau(\phi)$ は以下の通りに計算する。式(32)に式(29)を代入すると次式が得られる。

$$f_\tau(\phi) = \pm 2 \int_0^\phi \frac{\mu p \theta}{\phi} \sin \theta \frac{h}{2 \sin \phi} d\theta = \pm h \mu p \left(\frac{1}{\phi} - \frac{1}{\tan \phi} \right) \quad (33)$$

以上の計算をもとに $f(\phi)$ を求める。鉛直方向応力 p は位置角 ϕ のみの関数であるから

特に $p(\phi)$ と書くことにすれば、これと水平方向合力 $f(\phi)$ との関係式は以下の式により与えられる。

$$f(\phi) = f_i(\phi) + f_r(\phi) = h \left[p(\phi) \left\{ 1 \pm \mu \left(\frac{1}{\phi} - \frac{1}{\tan \phi} \right) \right\} - 2k\varpi(\phi, a) \right] \quad (34)$$

上式はロールと材料がすべり状態にある場合についての式であるが、特に固着摩擦の場合には $a=1$, $\varpi(\phi, 1) = \frac{\pi}{4}$ であるから、式(34)は式(35)の通りに変形できる。

$$f(\phi) = f_i(\phi) + f_r(\phi) = h \left[p(\phi) - 2k \left\{ \frac{\pi}{4} \mp \frac{1}{2} \left(\frac{1}{\phi} - \frac{1}{\tan \phi} \right) \right\} \right] \quad (35)$$

式(31)にて定義されている $\varpi(\phi, a)$ は接触角 ϕ の関数であるが、実際には接触角の影響は非常に小さい。たとえば、接触角の 30° の違いに対して、 $\varpi(\phi, a)$ の差が最大となる $a=1$ の場合でも差は 1% にすぎないので、数値計算の過程では ϖ は a のみの関数として取りあつかっても差し支えない。

2.4. 応力分布の計算

スラブ要素に関する圧延方向力（水平力）の釣合条件式(27)と、式(34)または式(35)とを組み合わせることにより、水平方向合力に関する微分方程式が得られる。この微分方程式は、

$$\frac{df(\phi)}{d\phi} = F(f(\phi), \phi) \quad (36)$$

なる形をしており、この 1 階常微分方程式の初期値問題は数値積分により解かねばならない。

次章にて説明する数値解析プログラムでは、Runge-Kutta-Gill 法により式(36)の解を求めているが、その概略は以下の通りである。初期値である、接触開始位置 $\phi = \phi_0$ およびロール下死点 $\phi = 0$ での水平方向合力 $f(\phi)$ は、後方張力 σ_b と前方張力 σ_f （通常の応力符号の定義通り引張りを正とする）をもとに、 $f(\phi_0) = -h_2 \cdot \sigma_b$, $f(0) = -h_1 \cdot \sigma_f$ と与えら

れる。積分領域である接触弧を N 等分し、分点を入口より出口に向けて $\phi_0, \phi_1, \phi_2, \dots, \phi_N$ とする。分点 n での水平方向合力 $f(\phi_n)$ の値が求められているものとする、分点 $n+1$ での水平方向合力の値 f_{n+1} は以下の近似式より求めることができる。

$$f_{n+1} \cong f_n + \frac{1}{3} \left(\frac{1}{2} \Delta f^{(0)} + \frac{3}{2} \Delta f^{(1)} + \frac{1}{2} \Delta f^{(2)} + \frac{1}{2} \Delta f^{(3)} \right) \quad (37)$$

$$\left. \begin{aligned} \Delta f^{(0)} &= F(f(\phi_n), \phi_n) \cdot \Delta \phi \\ \Delta f^{(1)} &= F\left(f(\phi_n) + \frac{1}{2} \Delta f^{(0)}, \phi_n + \frac{1}{2} \Delta \phi\right) \cdot \Delta \phi \\ \Delta f^{(2)} &= F\left(f(\phi_n) - \frac{1}{2} \Delta f^{(0)} + \Delta f^{(1)}, \phi_n + \frac{1}{2} \Delta \phi\right) \cdot \Delta \phi \\ \Delta f^{(3)} &= F\left(f(\phi_n) + \frac{1}{2} \Delta f^{(1)} + \frac{1}{2} \Delta f^{(2)}, \phi_n + \Delta \phi\right) \cdot \Delta \phi \end{aligned} \right\} \quad (38)$$

ただし $\Delta \phi$ は分点の幅であり、入口側（上流側）からの計算については $\Delta \phi = \phi_i - \phi_{i-1}$ 、出口側（下流側）からの計算については $\Delta \phi = \phi_{i-1} - \phi_i$ である。式(37)、式(38)により逐次各分点での水平方向合力の値が入口側および出口側より計算され、式(34)または式(35)に代入することにより、接触弧での圧延圧力分布 $p(\phi_n)$ が計算できる。式(37)、式(38)によ

り入口側、出口側より求めた $p(\phi_n)$ が一致する位置が中立面である。圧下力は圧延圧力を接触弧内で積分することにより求め、トルクは摩擦応力にロール半径を乗じ接触弧内で積分することにより求める。

2.5. すべり／固着の判定

摩擦応力はせん断降伏応力を超えることはできないので、 $\tau_f = \mu p$ より計算した値がせん断降伏応力を超える場合、摩擦応力＝せん断降伏応力という条件のもとで計算を実施する。式(31)中の $\varpi(\phi, a)$ は、Orowan による図示結果を次式にて近似できる。

$$\varpi(\phi, a) \cong \varpi(a) \cong 1 - \left(1 - \frac{\pi}{4}\right) a^2 \quad (39)$$

式(39)を式(34)に代入して a について解くと次式が得られる。

$$a = \frac{-q + \sqrt{q^2 + 4 \left(1 - \frac{\pi}{4}\right) \left(1 + \frac{f(\phi)}{2hk}\right)}}{2 \left(1 - \frac{\pi}{4}\right)} \quad (40)$$

ただし、

$$q \equiv \frac{1}{2\mu} \left\{ 1 \mp \mu \left(\frac{1}{\phi} - \frac{1}{\tan \phi} \right) \right\} \quad (41)$$

である。各分点 ϕ_n について求められた $f(\phi_n)$ を式(40)、(41)に代入すると a の値が定まるので、入口側、出口側よりまずは式(34)、(36)を組み合わせて計算を行い、 $a \geq 1$ となる分点を境にして固着摩擦状態についての式(35)と式(36)とを組み合わせて計算を行う。

3. Orowan の理論に基づく数値解析プログラム

3.1. 入力データ (21~39 行)

R : ロール半径

H1、H2 : 圧延前板厚 h_1 、圧延後板厚 h_2

SGB、SGF : 後方張力 σ_b 、前方張力 σ_f ($\text{kgf} \cdot \text{mm}^{-2}$) (引張りを正とする)

NKM : 2次元降伏応力 k の折れ線近似点数 (最大 50 点)

XK(I)、RK0(I) : 2次元降伏応力折れ線近似点 (I) の圧延方向位置 x および 2次元降伏応力 k の値 (出口面より入口面に向かって入力する、 $\text{kgf} \cdot \text{mm}^{-2}$)

NUM : 摩擦係数 μ の折れ線近似点数 (最大 50 点)

XU(I)、U0(I) : 摩擦係数折れ線近似点 (I) の圧延方向位置 x および摩擦係数 μ の値 (出口面より入口面に向かって入力する)

3.2. 出力データ

1) 計算結果-1 (336~351 行、SUBROUTINE PRINT1)

X : 圧延方向位置 x

O : 位置角 ϕ (deg.、プログラム中の変数名 : OS)

SL/ST : すべり / 固着識別子 (= 1 : すべり、= 2 : 固着、プログラム中の変数名 : NSST)

F : 水平方向合力 $f(\phi)$ ($\text{kgf} \cdot \text{mm}^{-1}$)

P : 鉛直方向応力 $p(\phi)$ (=圧延圧力、 $\text{kgf}\cdot\text{mm}^{-2}$ 、プログラム中の変数名 : PM)

P/K : $\frac{p(\phi)}{k}$ (プログラム中の変数名 : PMK)

SGX : 圧延方向応力 σ_{xx} ($\text{kgf}\cdot\text{mm}^{-2}$)

SGX/K : $-\frac{\sigma_{xx}}{k}$ (プログラム中の変数名 : SGXK)

TAU : 摩擦せん断応力 τ_f ($\text{kgf}\cdot\text{mm}^{-2}$)

H : 板厚 h (mm)

U : 摩擦係数 μ

RK : 2次元降伏応力 k ($\text{kgf}\cdot\text{mm}^{-2}$)

A : $a = \frac{2\mu p(\phi)}{k}$ の値

W : $\varpi(\phi, a)$ の値

(以上は、出口側からの計算結果・入口側からの計算結果双方が表示される)

2) 計算結果-2 (403行~459行、SUBROUTINE PRINT2)

R : 圧下率 (プログラム中の変数名 : RS)

XL : 接触弧長 (mm)

OM : 接触角 (deg.、プログラム中の変数名 : OMS)

XN : 中立点の x 座標 (mm)

ON : 中立角 (deg.)

F、P : 中立点での $f(\phi), p(\phi)$ の値 (プログラム中での変数名 : C1、C2)

PM : 平均圧延圧力 ($\text{kgf}\cdot\text{mm}^{-2}$)

P : 圧延荷重 ($\text{kgf}\cdot\text{mm}^{-1}$ 、プログラム中での変数名 : SUMP)

G : 上下ロール合計トルク ($\text{ton}\cdot\text{m}\cdot\text{m}^{-1}$ 、プログラム中での変数名 : SUMG)

A : トルクアーム (mm、プログラム中での変数名 : TAM)

3.3. プログラム各行の説明

1) メインプログラム

21~39 : データの読み込み

40~48 : 変数の初期化

49~116 : 出口面側からの $f(\phi), p(\phi)$ などの計算

- 51～62 : 変数の初期化
- 63 : 摩擦係数の計算
- 64 : 2次元降伏応力の計算
- 65～70 : すべり／固着の判定
- 71～84 : すべり状態に関する式(27)、式(34)、式(37)の計算
 - 83 : すべり状態に対する式(37)の計算
- 85～100 : 固着状態に関する式(27)、式(35)、式(37)の計算
 - 99 : 固着状態に対する式(37)の計算
- 101～107 : プリントアウト、諸量の記憶
 - (中立点位置・圧延荷重などの算出などに用いる)

117～191 : 入口面側からの $f(\phi)$, $p(\phi)$ などの計算

(内容は 49～116 行と同じなので省略)

192 : 中立点位置の検索・圧延荷重など計算およびプリントアウト

193～216 : フォーマット文

2) サブルーチン

217～262 (OROSLP、FNLSIP) : すべり状態に対する式(34)の計算

263～300 (OROSTK、FNSTIK) : 固着状態に対する式(34)の計算

301～318 (FRICTN) : 摩擦係数の計算

319～335 (DEFOM) : 2次元降伏応力の計算

336～351 (PRINT1) : プリントアウト

352～402 (SERCH) : 中立点位置の検索・圧延荷重など計算およびプリントアウト

3.4. プログラムリスト

```

001 C
002 C      Generalized program for OROWAN'S equation.
003 C
004 C      1973. 11. 12      H. Matsuno
005 C      1989.  7.  9      J. Yanagimoto
006 C
007      PROGRAM MAIN
008 C
009          COMMON /COM0 / XRV1 (500), XRV2 (500), ORV1 (500), ORV2 (500)
010          COMMON /COM1 / PRV1 (500), PRV2 (500), FRV1 (500), FRV2 (500)
011          COMMON /COM2 / TRV1 (500), TRV2 (500)
012          COMMON /COM3 / X, O, NSIO, NSST, U, RK, A, W, PAI, H
013          COMMON /COM4 / F, PM, PMK, SGX, SGXK, TAU, TAUk, PR
014          COMMON /COM5 / NUM, NKM, XU (50), UO (50), XK (50), RK0 (50)
015          COMMON /COM6 / NEXT, NENT, NT1, NT2, NTO
016          COMMON /COM7 / DH, H1, XL, OM, SUMP, SUMG, TAM
017 C
018 C ... Initialization.
019      PAI= 3.14159265
020      NPMAX= 10
021 C
022 C ... Data input.
```

```

023     WRITE(6, 1001)
024     READ(5, *) R
025     WRITE(6, 1002)
026     READ(5, *) H1, H2
027     WRITE(6, 1003)
028     READ(5, *) SGB, SGF
029     WRITE(6, 1004)
030     READ(5, *) NKM
031     DO 1 I=1, NKM
032     WRITE(6, 1005) I
033     1 READ(5, *) XK(I), RK0(I)
034     WRITE(6, 1006)
035     READ(5, *) NUM
036     DO 2 I=1, NUM
037     WRITE(6, 1007) I
038     2 READ(5, *) XU(I), U0(I)
039     WRITE(6, 1009) R, H1, H2, SGF, SGB
040 C
041 C ... Data Initialization.
042     DIA = 2.0*R
043     FF  =-SGF*H2
044     FB  =-SGB*H1
045     DH  = H1-H2
046     XL  = SQRT(R*DH-0.25*DH**2)
047     OM  = ATAN(XL/(R-0.5*(H1-H2)))
048     DQM = OM/400.
049 C
050 C ... Calculation from exit side.
051     WRITE(6, 1080)
052     WRITE(6, 2002)
053     WRITE(6, 1501)
054     DQ  = DQM
055     O   = 0.
056     F   = FF
057     FN  = F
058     NSIO= 2
059     NP  = NPMAX
060     DO 50 I=1, 500
061     X= R*SIN(O)
062     H= H2+ 2.0*R*(1.0-COS(O))
063     CALL FRICTN(X, U)
064     CALL DEFOM(X, RK)
065     C1= -0.5/U
066     C2= 1.0-0.25*PAI
067     C3= 1.0+F/H/RK
068     C4= SQRT(C1**2+4.0*C2*C3)
069     A  = 0.5*(C1+C4)/C2
070     IF(A.GE.1.0) GO TO 10
071     NSST= 1
072     W= -C2*A**2+1.0
073     IF(O.GT.0.0) GO TO 9
074     C2= 1.0
075     GO TO 8
076     9 CONTINUE
077     C1= SIN(O)/COS(O)
078     C2= 1.0+U*(1.0/O-1.0/C1)
079     8 CONTINUE
080     PM= (FN/H+RK*W)/C2
081     SGX= FN/H
082     TAU= U*PM
083     CALL OROSLP(F, FN, O, DQ, H, U, RK, W, DIA, NSIO)
084     GO TO 20
085     10 CONTINUE

```

```

086      NSST= 2
087      W= 0.25*PAI
088      C1= SIN(O)/COS(O)
089      IF(O.GT.0.0) GO TO 15
090      C2= W
091      GO TO 16
092 15 CONTINUE
093      C2= W-0.5*(1.0/O-1.0/C1)
094 16 CONTINUE
095      V= C2*SIN(O)+0.5*COS(O)
096      PM= FN/H+RK*C2
097      SGX= FN/H
098      TAU= 0.5*RK
099      CALL OROSTK(F, FN, O, DQ, H, U, RK, V, DIA, NSIO)
100 20 CONTINUE
101      PRV2(I)= PM
102      FRV2(I)= F
103      TRV2(I)= TAU
104      XRV2(I)= X
105      ORV2(I)= 0
106      IF(NP.LT.NPMAX) GO TO 30
107      CALL PRINT1
108      NP= 0
109 30 CONTINUE
110      F= FN
111      O= O+DQ
112      IF(O.GE.OM) GO TO 41
113      NP= NP+1
114 50 CONTINUE
115 41 CONTINUE
116      NEXT= I
117 C
118 C ... Calculation from entry side.
119      WRITE(6,2001)
120      WRITE(6,1501)
121      DQ = DQM
122      O = OM
123      F = FB
124      FN = F
125      NSIO= 1
126      NP = NPMAX
127      DO 150 I=1, 500
128      X= R*SIN(O)
129      H= H2+ 2.0*R*(1.0-COS(O))
130      CALL DEFOM(X, RK)
131      CALL FRICTN(X, U)
132      C1= -0.5/U
133      C2= 1.0-0.25*PAI
134      C3= 1.0+F/H/RK
135      C4= SQRT(C1**2+4.0*C2*C3)
136      A = 0.5*(C1+C4)/C2
137      IF(A.GE.1.0) GO TO 110
138      NSST= 1
139      W= -C2*A**2+1.0
140      IF(O.GT.0.0) GO TO 109
141      C2= 1.0
142      GO TO 108
143 109 CONTINUE
144      C1= SIN(O)/COS(O)
145      C2= 1.0+U*(1.0/O-1.0/C1)
146 108 CONTINUE
147      PM= (FN/H+RK*W)/C2
148      SGX= FN/H

```

```

149     TAU= U*PM
150     U= -U
151     DQ= -DQ
152     CALL OROSLP(F, FN, O, DQ, H, U, RK, W, DIA, NSIO)
153     DQ= -DQ
154     U= -U
155     GO TO 120
156 110 CONTINUE
157     NSST= 2
158     W= 0.25*PAI
159     C1= SIN(O)/COS(O)
160     IF(O.GT.0.0) GO TO 115
161     C2= W
162     GO TO 116
163 115 CONTINUE
164     C2= W+0.5*(1.0/O-1.0/C1)
165 116 CONTINUE
166     V= C2*SIN(O)-0.5*COS(O)
167     U= -U
168     DQ= -DQ
169     PM= FN/H+RK*C2
170     SGX= FN/H
171     TAU= 0.5*RK
172     CALL OROSTK(F, FN, O, DQ, H, U, RK, V, DIA, NSIO)
173     DQ= -DQ
174     U= -U
175 120 CONTINUE
176     PRV1(I)= PM
177     FRV1(I)= F
178     TRV1(I)= TAU
179     XRV1(I)= X
180     ORV1(I)= 0
181     IF(NP.LT.NPMAX) GO TO 130
182     CALL PRINT1
183     NP= 0
184 130 CONTINUE
185     F= FN
186     O= O-DQ
187     IF(O.LE.0.0) GO TO 141
188     NP= NP+ 1
189 150 CONTINUE
190 141 CONTINUE
191     NENT= I
192     CALL SERCH
193 C
194 C ... Format statements.
195 1001 FORMAT(///' *** Generalized program for OROWAN-S equation.***',
196 *      //' ... Enter roll radius.')
197 1002 FORMAT( //' ... Enter strip thickness before & after rolling.')
198 1003 FORMAT( //' ... Enter back & front tension of strip.')
199 1004 FORMAT( //' ... Enter number of intervals for yield stress.')
200 1005 FORMAT( ' ... Enter X-co. & yield stress for pnt.no.',I3)
201 1006 FORMAT( //' ... Enter number of intervals for fric. coeff.')
202 1007 FORMAT( ' ... Enter X-co. & fric. coeff. for pnt.no.',I3)
203 1009 FORMAT( //' *** Input data ***',
204 *      //' Roll radius R=',F10.4,' (mm)',
205 *      '/' Thickness of strip H1,H2=',2F10.4,' (mm)',
206 *      '/' Front and back tensions SGF,SGB=',2F10.4,
207 *      '(kgf/mm2)')
208 1080 FORMAT(///' *** CALCULATED RESULT -- 1 ***')
209 2001 FORMAT(///T5,' *** Stress distribution on entry side.***')
210 2002 FORMAT(///T5,' *** Stress distribution on exit side.***')
211 1501 FORMAT(T15,' X', T23,' O', T29,' SL/ST', T42,' F', T50,' P',

```

```

212      *      T57, 'P/K', T66, 'SGX', T72, 'SGX/K', T81, 'TAU',
213      *      T94, 'H', T102, 'U', T109, 'RK', T118, 'A', T126, 'W')
214 C
215      STOP
216      END
217 C
218 C  SUBR. OROSLP
219      SUBROUTINE OROSLP(F, FN, O, DQ, H, U, RK, W, DIA, NSIO)
220 C
221 C *** Subroutine for orowan slipping. ***
222 C
223      FS= F
224      OS= 0
225      CALL FNSLIP(FS, FNS, OS, H, U, RK, W, DIA, NSIO)
226      CO= DQ*FNS
227      OS= O+0.5*DQ
228      FS= F+0.5*CO
229      CALL FNSLIP(FS, FNS, OS, H, U, RK, W, DIA, NSIO)
230      C1= DQ*FNS
231      OS= O+0.5*DQ
232      FS= F-0.5*CO+C1
233      CALL FNSLIP(FS, FNS, OS, H, U, RK, W, DIA, NSIO)
234      C2= DQ*FNS
235      OS= O+1.0*DQ
236      FS= F+0.5*C1+0.5*C2
237      CALL FNSLIP(FS, FNS, OS, H, U, RK, W, DIA, NSIO)
238      C3= DQ*FNS
239      C4= 0.5*CO+1.5*C1+0.5*C2+0.5*C3
240      FN= F+C4/3.0
241      RETURN
242      END
243 C
244 C  SUBR. FNSLIP
245      SUBROUTINE FNSLIP(FS, FNS, OS, H, U, RK, W, DIA, NSIO)
246 C
247 C *** Subroutine for slipping function. ***
248 C
249      C1= SIN(OS)+U*COS(OS)
250      C2= FS*DIA*C1/H
251      C3= DIA*RK*W*C1
252      FNS= C2+C3
253      IF(OS.GT.0.0) GO TO 10
254      C2= 1.0
255      GO TO 11
256 10 CONTINUE
257      C1= SIN(OS)/COS(OS)
258      C2= 1.0+U*(1.0/OS-1.0/C1)
259 11 CONTINUE
260      FNS= FNS/C2
261      RETURN
262      END
263 C
264 C  SUBR. OROSTK
265      SUBROUTINE OROSTK(F, FN, O, DQ, H, U, RK, W, DIA, NSIO)
266 C
267 C *** Subroutine for orowan sticking. ***
268 C
269      FS= F
270      OS= 0
271      CALL FNSTIK(FS, FNS, OS, H, U, RK, W, DIA, NSIO)
272      CO= DQ*FNS
273      OS= O+0.5*DQ
274      FS= F+0.5*CO

```

```

275     CALL FNSTIK (FS, FNS, OS, H, U, RK, W, DIA, NSIO)
276     C1= DQ*FNS
277     OS= 0+0. 5*DQ
278     FS= F-0. 5*C0+C1
279     CALL FNSTIK (FS, FNS, OS, H, U, RK, W, DIA, NSIO)
280     C2= DQ*FNS
281     OS= 0+1. 0*DQ
282     FS= F+0. 5*C1+0. 5*C2
283     CALL FNSTIK (FS, FNS, OS, H, U, RK, W, DIA, NSIO)
284     C3= DQ*FNS
285     C4= 0. 5*C0+1. 5*C1+0. 5*C2+0. 5*C3
286     FN= F+C4/3. 0
287     RETURN
288     END
289 C
290 C   SUBR. FNSTIK
291     SUBROUTINE FNSTIK (FS, FNS, OS, H, U, RK, W, DIA, NSIO)
292 C
293 C *** Subroutine for sticking function. ***
294 C
295     C1= SIN(OS)
296     C2= FS*DIA*C1/H
297     C3= DIA*RK*W
298     FNS= C2+C3
299     RETURN
300     END
301 C
302 C   SUBR. FRICTN
303     SUBROUTINE FRICTN (X, U)
304 C
305         COMMON /COM5 / NUM, NKM, XU (50), U0 (50), XK (50), RK0 (50)
306 C
307     DO 10 I=1, NUM
308     U= 0. 1
309     IF (X. LT. XU (I )) GO TO 10
310     IF (X. GE. XU (I+1)) GO TO 10
311     C1= (U0 (I+1)-U0 (I)) / (XU (I+1)-XU (I))
312     U= U0 (I)+C1*(X-XU (I))
313     U= ABS (U)
314     GO TO 20
315 10 CONTINUE
316 20 CONTINUE
317     RETURN
318     END
319 C
320 C   SUBR. DEFOM
321     SUBROUTINE DEFOM (X, RK)
322 C
323         COMMON /COM5 / NUM, NKM, XU (50), U0 (50), XK (50), RK0 (50)
324 C
325     DO 10 I=1, NKM
326     RK= 20. 0
327     IF (X. LT. XK (I )) GO TO 10
328     IF (X. GE. XK (I+1)) GO TO 10
329     C1= (RK0 (I+1)-RK0 (I)) / (XK (I+1)-XK (I))
330     RK= RK0 (I)+C1*(X-XK (I))
331     GO TO 20
332 10 CONTINUE
333 20 CONTINUE
334     RETURN
335     END
336 C
337 C   SUBR. PRINT1

```

```

338     SUBROUTINE PRINT1
339 C
340         COMMON /COM3 / X, O, NSIO, NSST, U, RK, A, W, PAI, H
341         COMMON /COM4 / F, PM, PMK, SGX, SGXK, TAU, TAUK, PR
342 C
343         PMK= PM/RK
344         SGXK= SGX/RK
345         TAUK= TAU/RK
346         OS= 180.0*O/PAI
347         WRITE(6, 100) X, OS, NSST, F, PM, PMK, SGX, SGXK, TAU, H, U, RK, A, W
348 100    FORMAT(T10, 2F8. 2, I6, 5X, 2F8. 2, F8. 3, F8. 2, F8. 3, F8. 2,
349         *          4X, F8. 2, F8. 3, 2F8. 2, F8. 3)
350         RETURN
351         END
352 C
353 C     SUBR. SERCH
354     SUBROUTINE SERCH
355 C
356         COMMON /COM0 / XRV1(500), XRV2(500), ORV1(500), ORV2(500)
357         COMMON /COM1 / PRV1(500), PRV2(500), FRV1(500), FRV2(500)
358         COMMON /COM2 / TRV1(500), TRV2(500)
359         COMMON /COM4 / F, PM, PMK, SGX, SGXK, TAU, TAUK, PR
360         COMMON /COM6 / NEXT, NENT, NT1, NT2, NTO
361         COMMON /COM7 / DH, H1, XL, OM, SUMP, SUMG, TAM
362 C
363         NTO= 0
364         DO 10 I=1, NEXT
365             J= NENT-I+1
366             DF= FRV1(J)-FRV2(I)
367             IF (DF. GT. 0. 0) GO TO 10
368             GO TO 20
369 10    CONTINUE
370         NTO= 1
371         GO TO 30
372 20    CONTINUE
373         NT1= J
374         NT2= I
375         IF (NT2. NE. 1) GO TO 40
376         IF (DF. GE. -0. 01) GO TO 40
377         NTO= 1
378         GO TO 30
379 40    CONTINUE
380         NT1= NT1-1
381         SUMP1= 0. 0
382         SUMG1= 0. 0
383         IF (NT1. EQ. 0) GO TO 50
384         DO 60 I=1, NT1
385             SUMP1= SUMP1+PRV1(I)*(XRV1(I)-XRV1(I+1))
386             SUMG1= SUMG1+PRV1(I)*(XRV1(I)-XRV1(I+1))*XRV1(I)
387 60    CONTINUE
388 50    CONTINUE
389         SUMP2= 0. 0
390         SUMG2= 0. 0
391         DO 70 I=1, NT2
392             SUMP2= SUMP2+PRV2(I)*(XRV2(I+1)-XRV2(I))
393             SUMG2= SUMG2+PRV2(I)*(XRV2(I+1)-XRV2(I))*XRV2(I)
394 70    CONTINUE
395         SUMP= SUMP1+SUMP2
396         SUMG= 2. 0*(SUMG1+SUMG2)/1000. 0
397         TAM= 1000. 0*SUMG/(2. 0*SUMP)
398         PM= (SUMP1+SUMP2)/XL
399 30    CONTINUE
400         CALL PRINT2

```

```

401     RETURN
402     END
403 C
404 C   SUBR. PRINT2
405     SUBROUTINE PRINT2
406 C
407         COMMON /COM0 / XRV1(500),XRV2(500),ORV1(500),ORV2(500)
408         COMMON /COM1 / PRV1(500),PRV2(500),FRV1(500),FRV2(500)
409         COMMON /COM2 / TRV1(500),TRV2(500)
410         COMMON /COM3 / X, O, NSIO, NSST, U, RK, A, W, PAI, H
411         COMMON /COM4 / F, PM, PMK, SGX, SGXK, TAU, TAUk, PR
412         COMMON /COM6 / NEXT, NENT, NT1, NT2, NTO
413         COMMON /COM7 / DH, H1, XL, OM, SUMP, SUMG, TAM
414 C
415     RS= DH/H1
416     OMS= 180.0*OM/PAI
417     ON= ORV2(NT2)
418     ON= 180.0*ON/PAI
419     XN= XRV2(NT2)
420     WRITE(6,100)
421 100 FORMAT(///'*** CALCULATED RESULT -- 2 ***'///)
422     IF(NTO.EQ.1) GO TO 10
423     WRITE(6,110)
424 110 FORMAT(T10,'* REDUCTION RATIO')
425     WRITE(6,111) RS
426 111 FORMAT(T14,'R',F8.3,' (--)')
427     WRITE(6,120)
428 120 FORMAT(T10,'* CONTACT LENGTH')
429     WRITE(6,121) XL,OMS
430 121 FORMAT(T14,'XL',F8.2,' (MM)',4X,'OM',F8.2,' (DEGREE)')
431     WRITE(6,130)
432 130 FORMAT(T10,'* NEUTRAL POINT')
433     WRITE(6,131) XN,ON
434 131 FORMAT(T14,'XN',F8.2,' (MM)',4X,'ON',F8.2,' (DEGREE)')
435     C1= FRV2(NT2)
436     C2= PRV2(NT2)
437     WRITE(6,135)
438 135 FORMAT(T10,'* NEUTRAL POINT PRESSURE')
439     WRITE(6,136) C1,C2
440 136 FORMAT(T14,'F',F8.2,' (KG/MM)',4X,'P',F8.3,' (KG/MM2)')
441     WRITE(6,140)
442 140 FORMAT(T10,'* MEAN ROLL PRESSURE')
443     WRITE(6,141) PM
444 141 FORMAT(T14,'PM',F8.2,' (KG/MM2)')
445     WRITE(6,150)
446 150 FORMAT(T10,'* ROLLING FORCE AND TORQUE')
447     WRITE(6,151) SUMP,SUMG
448 151 FORMAT(T14,'P',F8.2,' (T/M)',4X,'G',F8.3,' (T-M/M)')
449     WRITE(6,160)
450 160 FORMAT(T10,'* TORQUE ARM')
451     WRITE(6,161) TAM
452 161 FORMAT(T14,'A',F8.3,' (MM)')
453     GO TO 20
454 10 CONTINUE
455     WRITE(6,170)
456 170 FORMAT(T10,'* NO NEUTRAL POINT')
457 20 CONTINUE
458     RETURN
459     END

```

3.5. 数値解析結果

図11に、ロール半径 400mm、圧延前後の板厚 $h_1 = 14.5\text{mm} \rightarrow h_2 = 9.5\text{mm}$ (熱延仕上げスタンド F 3 相当)、摩擦係数 $\mu = 0.3$ 、前後方張力 $\sigma_b = \sigma_f = 0.0$ の場合についての無次元化圧延圧力分布を示す。3次元剛塑性 FEM⁷⁾による2次元解析の結果求めた圧延圧力分布を同図中に示すが、Orowan の圧延理論により求めた圧延圧力分布は、剛塑性 FEM の結果と良く一致していることがわかる。熱延の様に摩擦係数が高く、さらにロール径に対して板厚が大きい条件については Karman の理論より Orowan の理論の方がより正しい解を与え、さらに剛塑性 FEM と比較しても遜色無い結果が得られる。なお、全域で摩擦応力 $\tau_f = k$ となった場合 Orowan の式は解析的に積分でき、これを Sims の式 (理論) と呼んでいる²²⁾。




図11 Orowan の理論による圧延圧力分布の剛塑性FEM解析結果との比較

4. 圧延での二次元変形と応力状態のまとめ・定常状態

4.1. ロールバイト内の変形と歪み分布

Karman の理論では板厚方向に均一な応力場を想定している。Orowan の理論では、板厚方向の応力分布があることを想定しており、この応力分布は板厚方向せん断応力に対応している。つまりせん断応力に対応して板厚方向応力分布が生じているわけである。冷間圧延の様に、十分な潤滑をして摩擦係数が低い場合にはせん断応力が小さいので、Karman の近似が利用できるのかもしれないが、実際には多少なりともせん断応力および変形が発生するはずであり、熱間圧延の様に摩擦係数が高い場合にはせん断変形がさらに大きくなる。

4.2. 圧延時の変形

図11-2に、圧延による変形を図示する。圧延によって板厚方向に不均一な変形が

現れ、圧延前に垂直であった線がたわむ。ロールと接触している面では材料が先に進む傾向があるが、これは中立点までの後進域において、ロールより素材には引き込み方向の力が加わり、この力の影響がせん断変形として残っているためである。図 1 1 - 2 右には、ひずみ分布を示す。

圧延方向に等間隔 \bar{L}_2 のスリット線を入れておいたとする。圧延前に垂直であったスリット線は圧延後にはたわむが、その間隔を板厚方向に見てみる。




図 1 1 - 2 圧延における材料の変形と圧延後のひずみ

圧延時の被圧延材は、先後端の一部の領域を除き、定常状態 (Steady state) となる。つまり、圧延時の応力 $\sigma(\mathbf{x}, t)$ 、速度を $\dot{\mathbf{u}}(\mathbf{x}, t)$ とした時に、定常状態では、

$$\sigma(\mathbf{x}, t) \approx \sigma(\mathbf{x})$$


$$\dot{\mathbf{u}}(\mathbf{x}, t) = \dot{\mathbf{u}}(\mathbf{x})$$

となり、時間 t の影響を受けず、場所 \mathbf{x} のみの関数となる。

この定常状態では、圧延方向ひずみ (延伸) $\varepsilon_l = \ln \frac{\bar{L}_1}{\bar{L}_2}$ は板厚方向に一様である。延伸の一様性は定常状態の特徴であり、板幅方向についても延伸は一様になる。定常状態における延伸の一様性が、圧延における板プロフィルの形成に重要な役割を果たす。また体積一定の条件から、板厚ひずみも板厚方向に一様となる。せん断ひずみは板厚方向に単調に分布しており、板厚中心でゼロであり、ロールとの接触面で最大値をとる。

4.3. ロールバイト内の応力分布と圧延圧力分布

応力分布は先に述べた Karman もしくは Orowan の圧延理論によって与えられる。得られる応力を模式的示したものが、**図 1 1 - 3** である。なお、図 1 1 - 3 はせん断変形の影響を無視した場合について示してあり、この場合の応力分布は板厚方向に一様になる。実際にはせん断応力が存在しているので、その影響により応力の分布は板厚方向に一様にはならない。この正しい応力場は、Orowan の圧延理論によって計算できる。



ロールバイト入口、ロールバイト出口での圧延方向応力は、後方張力もしくは前方張力に等しい。中立点に至るまで圧延圧力は単調に増加するが、この理由は平板による圧縮と同じく摩擦によっており、同じくフリクション・ヒルと呼ばれる。フリクション・ヒルの高さは、摩擦係数が高いほど、接触長に較べて板厚が小さいほど、高くなる。

後方張力もしくは前方張力が作用すると、圧延圧力の大きさは減少する。従って圧延荷重は減少する。二次元理論で論じている圧延荷重は単位幅（紙面奥行き方向が幅方向）あたりの荷重であり、この荷重を単位幅圧延荷重、あるいは、線荷重と呼ぶ。また、中立点の位置は前方張力もしくは後方張力の値に応じて変動する。例えば後方張力が作用すると、無張力時と比較して中立点が出口側に移動する。

5. 板圧延における3次元変形と板プロフィルの形成機構

5.1. 圧延の3次元解析の分類

圧延における3次元変形を解明する動機は2つある。一つは条鋼系（棒鋼・線材・形鋼）の様に、幅広がり代表されるC断面（圧延方向横断面）内の変形が主体である場合である。この分野の初等解析法に基づく3次元圧延理論は、板幅/板厚が10を下回

る範囲での幅広がり の 解明 ¹⁴⁾以後 精力的に 研究が 進められたが、 現在では 有限要素法による 3次元解析が 主流となっており ⁷⁾、 **図12-1~3**に 示す様な 3次元変形解析の結果が、 パソコン上での 解析システムにより ごく短時間（棒線材圧延の場合には、 10~30分/1パス）で 計算できるようになっている。




図12-1 3次元FEMによる棒線材圧延加工の計算結果




図12-2 棒線材圧延3次元 FEM解析システム




図12-3 非対称アングル材圧延の計算結果

一方、薄板・厚板圧延の場合には、多段ロール系と被圧延材との相互作用により引き起こされる板厚の板幅方向分布の解明が重要な役割を担っている。これを解くためには、被圧延材の圧延理論を3次元化すると同時に、多段ロール系の弾性変形を連成して解く必要がある。多段ロール系の弾性変形の解析は、Shohet による分割モデルの提示に始まり ¹⁸⁾、初等解析法による被圧延の3次元変形と組み合わせた板プロフィール解析 ¹⁵⁾へと発展した。本章では、クラウン・エッジドロップの形成機構と、この解析手法について、ロール系の弾性変形を主体に概要を述べる。

5.2. 板プロフィール（板クラウン）の形成機構

板圧延後の板プロフィールは、材料の塑性変形と、圧延機・ロールの弾性変形との相互作用の結果として決まる。さらに圧延機・ロールの弾性変形は、4段圧延機を例に取る

と以下の要素に分解できる。




図 1 2 - 4 圧延機ハウジングの変形とロールの変形

- ワークロール～材料界面でのワークロールの扁平変形
- ワークロールのたわみ変形
- ワークロール～バックアップロール界面での両ロールの扁平変形
- バックアップロールのたわみ変形
- 圧延機ハウジングの変形

図 1 2 - 4 に、圧延機ハウジングの変形とワークロールたわみ変形を模式的に示す。圧延後板プロフィールは、上に述べた圧延機・ロールの弾性変形と材料の塑性変形との相互作用の結果として定まる。なお、ワークロールのたわみ変形、ワークロール～バックアップロール界面での両ロールの扁平変形、バックアップロールのたわみ変形については、材料力学すなわち円柱梁のたわみ理論を基礎とする分割モデルと 2 円筒近接式により解くことができる。

被圧延材（板）プロフィールは、中央部分での板厚差である**板クラウン**と、板端（エッジ）部での板厚の急激な減少である**エッジドロップ**に区別して論じる。板クラウン $C_h \equiv h_{1,c}$ は、板幅中心での板厚と板最エッジから 20mm 地点での板厚差により、次式により定義される²⁴⁾。

$$C_h \equiv h_{1,c} - h_{1,20} \quad (59)$$

ここで $h_{1,c}$ は圧延後の板幅中心での板厚、 $h_{1,20}$ は板最エッジから20mm地点での板厚である。被圧延材の入口（圧延前）の板クラウンは、

$$C_{h_2} \equiv h_{2,c} - h_{2,20} \quad (60)$$

である。**板クラウン比率**は板幅中心部の板厚ひずみと板最エッジ部から20mm地点での板厚ひずみの偏差であり、圧延前については $\frac{C_{h_2}}{h_2}$ 、圧延後については $\frac{C_{h_1}}{h_1}$ である。

被圧延材の初期板厚が均一であったとしても、ロールの弾性変形に伴い図12-5のような板プロフィールを取る傾向にあると考えられる。そのため板厚ひずみは、板中央部で最も小さな圧縮ひずみに、板エッジで最も大きな圧縮ひずみを取る。つまり、板幅方向中心よりエッジに向かって、板厚ひずみの大きさは増加する。




図12-5 板プロフィールとひずみ、張力分布

また、板幅方向ひずみは、板最エッジ部自由端の影響により板端部で有意な値を取る。すなわち板最エッジ部で被圧延材は3次元変形をする。体積一定条件より、板厚ひずみの絶対値は、圧延方向ひずみ（延伸）と板幅ひずみの和の絶対値に等しいから、熱延の

様に板厚が比較的厚い場合には、図 1 2 - 5 中に示したとおり圧延方向ひずみ（延伸）はクォーター部付近で最大となる傾向にあると考えることができる。

被圧延材の圧下率は板幅中央で小さく、板幅端部で大きな値を取り、材料が自由に圧延方向に伸びることができるとする。簡単のため、幅方向の変形を無視して考える。すると、塑性変形の体積一定条件から、板幅中央部では伸びが小さく、板幅端部では伸びが大きいことになる。図 1 2 - 7 に、ロール変形に応じて板クラウンが形成される場合の、幅方向の流れが無い場合に、圧延方向に起こりえる伸びを模式的に示す。




図 1 2 - 7 圧延方向に起こりえる伸び

ところが圧延では、先に述べたとおり、板幅方向に見た圧延方向ひずみ分布はほぼ均一であり、伸びの差は材料の弾性変形範囲でしか許容されない。従って、板幅方向に見た圧下率差に起因する圧延方向伸び差を解消する方向に、変形域入口面・出口面で幅方向に張力分布が発生する。その結果、圧延方向ひずみが大きい部分には圧縮側応力が、小さい部分には引張り側応力が発生し、この応力（張力分布）により、圧延方向ひずみの幅方向に見た均一性が保たれる。

以下に、図 1 2 - 8 を用いて、材料の塑性変形とロール、圧延機の相互作用について考察する。先に述べた通り、ロールには、ワークロールの扁平変形、ワークロールのたわみ変形、ワークロール～バックアップロール界面での両ロールの扁平変形、バックアップロールのたわみ変形が重畳された結果、入口材料厚さが幅方向に均一で、ワークロール、バックアップロールの半径がロール胴長方向に均一である場合にも、圧延後材料は幅方向に厚さ分布を持つ傾向にある。すなわち、板幅中央での板厚が厚く、板幅端部での板厚が薄くなり、板幅中央ではロール変形が大きくなりがちであり、圧下率小、伸び小であるため圧延方向張力が正の方向に大（引張り）となる。一方、板幅端部ではロール変形が小さくなりがちであるため、圧下率大、伸び大であるため負の方向に大（圧縮）の圧延方向張力が発生する。




図 1 2 - 8 張力フィードバック機構

ところが2次元圧延理論により示されている通り、変形域入出口面での張力が引張り側に大きいほど圧延圧力は減少し、圧縮側に大きいほど増加する。

板幅中央では、伸び小であるため圧延方向張力が正の方向に大（引張り）となる。そのため圧延圧力、圧延荷重（線荷重）が減少する。板幅端部では、負の方向に大（圧縮）の圧延方向張力が発生するため、この部分での圧延圧力、圧延荷重（線荷重）が増加する。つまり、板幅中央で圧延荷重が減少、板端部で圧延荷重が増加し、ロール変形の結果生じると思われる板クラウンを小さくする作用がある。この作用は、定常状態であるがゆえに発生する板面内の圧延方向張力に起因しており、**まさに、板圧延の物理である、** **といて良い。この、ロールの弾性変形の影響を緩和する方向に働く力（メカニズム）が、薄板の圧延を成立させている大きな要因である。**

以上は板中央部と板端部を代表例として示したが、板幅方向の任意の位置についても成立している。ある位置で圧延荷重が上がり、ロール変形が大きくなったとする。この大きなロール変形圧下率を減少させ、（不足する延伸を補おうとするために）板面内の張力を引張り側に増加させる、その結果圧延荷重は減少し、ロール変形を小さくしようとする。つまり、ロール変形が大きくなろうとしても、材料側からはロール変形をむしろ小さくしようとする作用が働き、結局圧延後材料の板厚分布を相殺することになる。こ

の機構を、**張力のフィードバック機構**と呼ぶ。図12-8（下）に張力のフィードバック機構をフローチャートとして示す。




図12-6 板圧延時の板プロフィールとロールの弾性変形との関係

5.3. 板幅方向流れの影響とエッジドロップ

薄板の場合には板のエッジ近傍では、幅方向流れを拘束する力学的作用は無いので、板幅方向流れが発生する。今までは板幅方向流れを無いものとして説明してきたが、現実には図12-5に示した通り板のエッジには幅方向の流れが存在するため、延伸が不足しこの部分の張力が引張側に上昇し、圧延荷重が下がりロール変形が大きくなるため、板端部では板厚が急激に減少する。**板の最エッジ部での流れに伴う板厚の急激な減少をエッジドロップと呼ぶ。**

図12-6に、弾性変形の影響による板プロフィールの変化を、ワークロール～材料界面での扁平変形の影響と、たわみ変形の影響に分けて示す。ワークロールのたわみ変形は二次曲線で近似でき板幅方向への変化も単調である。ワークロール～材料界面での扁平変形は、板端部で急激に大きくなる。これらの要因によって定まる板のプロフィールは、板幅中心部分寄りのボディ領域では緩やかに変化し（クラウン）、端部のエッジ領域では先に述べたとおり急激な変化をするが（エッジドロップ）、大まかに言えばボディ領域は板幅方向塑性流れが小さい領域、エッジ領域は板幅方向塑性流れが大きい領域である。板形状（平坦度）の変化があったとしても圧延方向ひずみ差への寄与は小さく、圧延方向ひずみは板厚方向に加え板幅方向にほぼ様な値をとる。板プロフィールを決める大部分の領域（ボディ部分、最エッジ部以外）では板幅方向の各位置で、（板厚方向ひずみ）+（板幅方向ひずみ）は一定の値をとる。

板のプロフィール変化は板厚方向ひずみの板幅方向への変化に対応しているので、板のプロフィールを変更するためには板幅方向のひずみが発生し、さらに板幅方向に分布している必要がある。板幅方向の塑性流れは板幅比が小さいほど起こりやすいので、板プロフィール、特にボディ部分でのクラウンの変更は板幅比が小さいスタンド、つまり板厚が大きいスタンドで行う方が望ましい。




図13 製品板厚が決まるメカニズム

6. ロール変形の解析手法

6.1. 分割モデルによるロール軸心たわみの解析

4段圧延機では、既に概略を述べたとおり、1)圧延機ハウジングの変形、2)バックアップロールのたわみ変形、3)ワークロール～バックアップロール間の接触扁平変形、4)ワークロールのたわみ変形、5)被圧延材～ワークロール間の接触扁平変形、6)ロールの熱膨張、7)被圧延材の3次元塑性変形が起こり、これらが相互に作用しあった結果製品の板厚が決まる。1)は狭義のミルストレッチ（ミル剛性）、1)～5)は広義のミルストレッチ（ミル剛性）である。これらの要因のうち、2)～5)が板幅方向板プロフィールに影響するロール系の弾性変形である。広義のミルストレッチと、被圧延材の塑性変形により板厚が決定されるメカニズムを図13に示す。ミルストレッチ線の傾きがミル剛性である。

ワークロール～バックアップロール系よりなる図14の系について、上記2)～4)を求めるための解析手法について説明する。4段圧延機の上1/4部分を幅方向に m 個の短冊要素に分割する。分割区間の幅を Δz 、区間 j について被圧延材より作用している単位幅あたりの圧延荷重（これを線荷重と呼ぶ）を $p(j)$ 、ワークロール～バックアップロール界面で作用している単位幅荷重を $q(j)$ 、各分割区間のバックアップロールたわみ量を $Y_b(i)$ 、ワークロールたわみ量を $Y_w(i)$ 、ショックあたりのワークロールベンディング力を J とする。さらに図14中には示していないが、ワークロールとバックアップロールの軸心近接量を ΔY_{BW} 、ワークロール～バックアップロール間のバネ定数を K 、半径あたりのワークロールクラウン、バックアップロールクラウンをそれぞれ $R_{CW}(i)$ 、 $R_{CB}(i)$ （凸クラウン側正）とする。なお、ワークロール～バックアップロール間のばね定数 K は、かなり広い条件範囲で $1/K = 3.1 \times 10^{-4} [\text{mm}/(\text{kg}/\text{mm})]$ 程度の値をとる。




図14 分割モデルによるロールたわみの計算手法

被圧延材より作用している単位幅あたりの圧延荷重 $p(j)$ は、被圧延材の塑性変形についての解析で求まり、対応してワークロール、バックアップロールに発生する変形は、梁理論に基づく以下の連立方程式を解くことにより得られる。

荷重の釣り合い式：

$$\sum_{j=1}^m q(j)\Delta z = \sum_{j=1}^m p(j)\Delta z + J \quad (42)$$

変位の適合条件式：

$$Y_B(i) - Y_W(i) + \Delta Y_{BW} + \frac{q(i)}{K} = R_{CW}(i) + R_{CB}(i) \quad (43)$$

さらに各区間の軸心たわみは、単純支持円柱梁理論により単位幅荷重と関連づけられる。 j 番目の分割区間に単位集中荷重を与えたときの i 番目の分割区間のたわみを、ワークロールは $\alpha_w(i, j)$ 、バックアップロールは $\alpha_b(i, j)$ であるとすれば、

ワークロールの変位～荷重関係式：

$$Y_w(i) + \sum_{j=1}^m \alpha_w(i, j)q(j)\Delta z = \sum_{j=1}^m \alpha_w(i, j)p(j)\Delta z \quad (44)$$

バックアップロールの変位～荷重関係式：

$$Y_B(i) - \sum_{j=1}^m \alpha_B(i, j)q(j)\Delta z = 0 \quad (45)$$

である。 $\alpha(i, j)$ は影響係数とも呼ばれており、次式により与えられる。

j が i の内側にある場合：

$$\alpha(i, j) = \frac{32}{3\pi E} \left[(1+\nu) \left\{ \frac{l}{d^2} + \frac{1}{D^2}(\beta - l) \right\} + \frac{2l^3}{d^4} + \frac{1}{D^4} (3\eta\beta L - 2l^3 - 3\beta^2\eta - \beta^3) \right] \quad (46)$$

j が i の外側にある場合：

$$\alpha(i, j) = \frac{32}{3\pi E} \left[(1+\nu) \left\{ \frac{l}{d^2} + \frac{1}{D^2}(\eta - l) \right\} + \frac{2l^3}{d^4} + \frac{1}{D^4} (3\eta\beta L - 2l^3 - 3\beta^2\eta - \eta^3) \right] \quad (47)$$

ただし、 $\eta \equiv \frac{L}{2} - (j - 0.5)\Delta z$ 、 $\beta \equiv \frac{L}{2} - (i - 0.5)\Delta z$ であり、(46)式、(47)式にそれぞれワークロールおよびバックアップロールの寸法を代入することによりそれぞれのロールについての $\alpha_w(i, j)$ 、 $\alpha_B(i, j)$ が計算できる。(42)式～(45)式により構成される合計 $3m+1$ この方程式を、バックアップロール界面で作用している単位幅荷重 $q(j)$ 、各分割区間のバックアップロールたわみ量 $Y_B(i)$ 、ワークロールたわみ量 $Y_w(i)$ 、ワークロールとバックアップロールの軸心近接量を ΔY_{BW} の合計 $3m+1$ 個の変数について解くことにより、被圧延材より作用している単位幅あたりの圧延荷重の分布 $p(j)$ に対応したロールたわみが得られる。

6.2. 被圧延材～ワークロール間の扁平変形

被圧延材より作用する圧延圧力 p により、ワークロールは弾性的に扁平することが知られている。板圧延の場合、接触弧長はロール半径に比較して十分に大きいので、半無限体に集中荷重が作用した場合の表面変位を重ねてワークロール表面の扁平変形を近似することが一般的に行われている¹⁹⁾。すなわち、

$$u_y(x, z) = \frac{1-\nu^2}{\pi E} \int_S \frac{p(\varphi, \psi)}{\sqrt{(x-\varphi)^2 + (z-\psi)^2}} d\varphi d\psi \quad (48)$$

により、圧延圧力分布 p を接触領域 S について積分し、扁平変位 u_y を求める。これをワークロールたわみと重ねることにより、弾性変形により引き起こされる幅方向ロールギャップ分布を知ることができる。ただし現実には、板幅はロール幅に比較して有限の大きさを持っているため、(48)式に有限幅の補正を加えた式により扁平変形を評価することが一般的に行われる²⁰⁾。




図15 ワークロールの扁平変形

なお通常の圧延条件では、扁平変形後のロールプロフィールが円弧を保つとしても解析精度を損なわないことが知られており（図15参照）、次式の Hitchcock によるロール扁平式が利用されることが多い。

$$R' = \left(1 + \frac{C^*}{h_2 - h_1} p^* \right) R \quad (49)$$

ただし $C^* = \frac{16(1-\nu^2)}{\pi E}$ 、 p^* は単位幅圧延荷重（圧延圧力を接触弧に沿って積分したもの）である。

6.3. 被圧延材の3次元変形の特徴とスリットモデル

被圧延材の3次元変形理論についての詳しい説明は省略するが、3次元変形理論において重要なのは、1)被圧延材の幅方向にみた圧延方向ひずみが等しい、という定常変形である圧延加工時の被圧延材の変形が当然満足すべき条件を、2)ロールの弾性変形に伴

うロールギャップの幅方向分布、と連成させて如何にして正しい理論を構築するのかといった点にある。既に述べたとおり、被圧延材（板）プロフィールは、ロールの弾性変形に伴い図12-5の様な‘板クラウン’を呈する。従って板厚ひずみは、一般に板幅方向中心よりエッジに向かって増加する。また、板幅方向ひずみは、板最エッジ部自由端の影響により板端部で有意な値を取る。すなわち板最エッジ部で被圧延材は3次元変形をする。体積一定条件より、板厚ひずみの絶対値は、圧延方向ひずみ（延伸）と板幅ひずみの和の絶対値に等しいから、熱延の様に板厚が比較的厚い場合には、図12-5中に示したとおり圧延方向ひずみ（延伸）はクォーター部付近で最大となる。

ところが圧延では、板幅方向に見た圧延方向ひずみ分布は均一であるから、圧延方向伸び差を解消する方向に、変形域入口面・出口面で張力分布が発生する。この機構を特に張力のフィードバック機構と呼ぶ。その結果、圧延方向ひずみが大きい部分には圧縮側応力が、小さい部分には引張り側応力が発生し、この応力（張力分布）により、圧延方向ひずみの幅方向に見た均一性が保たれる。

今、板を幅に短冊状の要素（スリット要素：図12-7）に分割し、それぞれの短冊要素の長さがすべて L_2 で等しかったものとする。圧延後の各短冊要素 i の長さを $L_1(i)$ とすれば、体積一定の条件から、

$$L_2 h_2 = L_1(i) h_1(i) \quad (50)$$

が成り立つ。ただし $h_1(i)$ は各短冊要素の圧延後板厚であり、(50)式には圧延前板厚が均一で h_2 であったことが含まれている（これはあくまでも話を簡単にするための前提である）。合計前方張力を T_f 、平均張力を $\bar{\sigma}_f$ とすれば、力の釣り合いから、

$$\sum_i \sigma_f(i) h_1(i) \Delta z = T_f = \bar{\sigma}_f \sum_i h_1(i) \Delta z \quad (51)$$

である。

張力のフィードバックにより付加される圧延方向ひずみが弾性範囲であるとするれば、

$$\varepsilon_1^E(i) = \frac{\sigma_f(i) - \bar{\sigma}_f}{E} \quad (52)$$

である。この付加的圧延方向ひずみは、材料の伸び差に大きさが等しく符号が反対である。すなわち、伸び差が（+）であるところには（-）の付加的圧延方向ひずみが発生し、圧延方向ひずみを均一に保とうとする。従って、

$$\varepsilon_1^E(i) = -\left[\ln\left(\frac{L_1(i)}{L_2}\right) - \ln\left(\frac{\bar{L}_1}{L_2}\right) \right] = \ln\left(\frac{\bar{L}_1}{L_1(i)}\right) \approx \frac{\bar{L}_1 - L_1(i)}{L_1(i)} = \frac{\bar{L}_1}{L_1(i)} - 1 = \frac{h_1(i)}{\bar{h}_1} - 1 \quad (53)$$

が成立し、張力分布は(52)式と(53)式より $\varepsilon_1^E(i)$ を消去することにより、

$$\sigma_f(i) = E\left(\frac{h_1(i)}{\bar{h}_1} - 1\right) + \bar{\sigma}_f \quad (54)$$

となる。ただし \bar{L}_1 は、付加的圧延方向ひずみが 0 である位置、すなわち、張力フィードバックを考慮した結果生じる前方張力 $(\sigma_f(i) - \bar{\sigma}_f)$ がゼロとなる位置の、短冊要素長さである。またこの位置での出口板厚を \bar{h}_1 とすれば、(50)式より導かれる関係式 $L_2 h_2 = L_1(i) h_1(i) = \bar{L}_1 \bar{h}_1$ が成立するが、(53)式を導く際にはこれを利用している。 \bar{h}_1 については、(54)式を(51)式を代入することにより、

$$\sum_i \left(\frac{h_1(i)}{\bar{h}_1} - 1\right) h_1(i) \Delta z = 0 \quad (55)$$

が満足される必要があるから、これを解くことにより次式の通りとなる。

$$\bar{h}_1 = \frac{\sum_i [h_1(i)]^2 \Delta z}{\sum_i h_1(i) \Delta z} \quad (56)$$

従って張力分布は、

$$\sigma_f(i) = E\left(h_1(i) \frac{\sum_i h_1(i) \Delta z}{\sum_i [h_1(i)]^2 \Delta z} - 1\right) + \bar{\sigma}_f \quad (57)$$

である。

解析は、2次元圧延理論 (Karman : 式(24)、Orowan : 式(36)) により求めた単位幅圧延荷重をもとに、まず式(42)~(45)の分割モデルと式(49)の扁平式を利用してロールギャップ $h_1(i)$ の分布を計算し、(57)式で求めた前方張力分布 $\sigma_f(i)$ をもとにさらに後方張力

分布 $\sigma_b(i)$ が前方張力分布と相似であるものとして(この近似はおおむね正しいことが証明されている)、再度2次元圧延理論により単位幅圧延荷重を計算し、ロールギャップ $h_1(i)$ の分布が収束するまでこの手順を繰り返す。(57)式を解いて得られる前方張力は、一般の圧延条件では中心部引張り、エッジ部圧縮となる。熱間圧延の様に比較的板厚が厚い条件では、図12-5に示した板幅方向流れが無視できないため、張力分布はエッジ部で引張りとなる。エッジ自由端の影響により引き起こされる被圧延材の幅方向流れを如何にして表現するのかという問題に直面し、さらにこの様なことを考慮するためには、Karman、Orowanによって与えられた2次元理論をそのまま利用することができない。

3次元変形理論としては各種のものが提示されているので、直接こちらを参照して頂くか¹⁵⁾、他のテキストをご参照頂きたい²¹⁾。

6.4. 平坦度と板クラウン比率変化

5.2 節で定めた板クラウン

$$C_{h_1} \equiv h_{1,C} - h_{1,20} \quad (59)$$

$$C_{h_2} \equiv h_{2,C} - h_{2,20} \quad (60)$$

と板クラウン比率を用いると、板クラウン(伸び差)と板形状(平坦度)の関係を得ることができる。圧延前後の板クラウン比率が同じである場合、つまり $\frac{C_{h_2}}{h_2} = \frac{C_{h_1}}{h_1}$ である場合には板幅方向中心と板最エッジ部との板厚ひずみ差がないということなので伸びひずみ差も無いことになり(なぜなら板幅方向中心から板エッジから20mmのボディ部では幅流れが無いと考えてよいので)、板は平坦のまま圧延される。 $\frac{C_{h_2}}{h_2} > \frac{C_{h_1}}{h_1}$ の場合には最エッジから20mm地点での板厚ひずみ差が小さいので中伸び、 $\frac{C_{h_2}}{h_2} < \frac{C_{h_1}}{h_1}$ の時は端伸びになり、伸び差が弾性範囲を超えると、**図16**に示す端伸びの形状が現れる。

板厚ひずみ差を $\Delta e_h \equiv \frac{C_{h_2}}{h_{2,C}} - \frac{C_{h_1}}{h_{1,C}}$ 、伸び差を Δe_l とする。体積一定の条件が

$$\Delta e_h + \frac{h_{2,20}}{h_{2,C}} \Delta e_l = 0 \text{ で表されるとすると、}$$

$$\Delta e_l = \frac{\frac{C_{h_1} - C_{h_2}}{h_{1,C}} - \frac{C_{h_1} - C_{h_2}}{h_{2,C}}}{\frac{h_{2,20}}{h_{2,C}}} = \frac{\frac{C_{h_1} - C_{h_2}}{h_{1,C}} - \frac{C_{h_1} - C_{h_2}}{h_{2,C}}}{1 - C_{h_2}} \quad (61)$$

により、板クラウン C_{h_2}, C_{h_1} と板クラウン比率 $\frac{C_{h_2}}{h_2}, \frac{C_{h_1}}{h_1}$ を伸び差に関連付けることができる。

6.5. 形状変化係数と平坦度不良

2次元理論の範囲では幅流れを無視しているのだから、板厚ひずみ偏差 $\Delta \varepsilon_h$ （基準位置からみた板厚ひずみ差）と伸びひずみ偏差 $\Delta \varepsilon_l$ との間に、 $\Delta \varepsilon_h = -\Delta \varepsilon_l$ の関係が満足される。ここで、形状変化係数 ξ を、板厚ひずみの偏差が形状に影響する度合いとして、板厚ひずみ偏差 $\Delta \varepsilon_h$ （基準位置からみた板厚ひずみ差）と伸びひずみ偏差 $\Delta \varepsilon_l$ （ Δe_l ）から定義する。

$$\xi = -\frac{\Delta \varepsilon_l}{\Delta \varepsilon_h} \quad (58)$$

この ξ を、有限要素法等による3次元理論により計算することが行われる。形状変化係数は、厚板・薄板圧延仕上げ全段までのクラウンの変更が容易な範囲で小さく、薄板圧延仕上げ後段や冷延のクラウンの変更が困難な範囲で1に近い値をとる。

薄板圧延では、板プロファイルの制御に伴い製品の平坦度（形状）が悪化する現象がしばしば観察される。特に後段側のスタンドでは板が薄いため板の座屈剛性が低く、同時に形状変化係数が大きいため板のプロファイルを変えようとするとき幅方向張力が大きく変動し、**図16**に示したような平坦度不良現象が現れやすくなる。この平坦度不良現象の存在が、後段スタンドでの板プロファイルの変更をさらに困難なものとしているが、この現象は圧延における3次元変形と深く関係しているわけである。

板の平坦度不良を表す急峻度（平坦度不良の周期と振幅の比） Λ は次式(62)で表される。

$$\Lambda \equiv \frac{\phi}{l} \quad (62)$$

急峻度と伸びひずみ差 $\Delta \varepsilon_l$ との関係は、式(63)の通りに導くことができる。

$$\Delta e_l \approx \frac{\Delta l}{l} = \left(\frac{\pi}{2}\right)^2 \Lambda^2 \quad (63)$$

平坦度が不良であると判断される急峻度 $\Lambda = 0.01$ は、式(63)に従い伸び差に換算すると 2×10^{-4} 程度の数値となり、弾性限界ひずみを下回るオーダーになる。このことが、平坦度不良の計算を困難としている要因である。今までに平坦度不良の予測に成功している研究は、初等理論を利用した文献¹⁵⁾以外には無いと思われ、ここに挑戦的な課題が横たわっている。




図16 平坦度不良（端伸び）

7. まとめ

圧延理論についてまとめた。さらに詳しく学習したい人は、板圧延に関連するテキスト²²⁾²³⁾を是非ご覧頂きたい。

参考文献(圧延理論共通)

- 1) Karman, T.: *Z. Math. Mech.*, **5**(1925), 139.
- 2) Nadai, A.: *J. Appl. Mech.*, **6**(1939), A54.
- 3) Bland, D.A. and Ford, H.: *Proc. Inst. Mech. Engr.*, **159**(1948), 144.
- 4) Orowan, E.: *Proc. Inst. Mech. Engr.*, **150**(1943), 140.
- 5) Sims, R.B.: *Proc. Inst. Mech. Engr.*, **168**(1954), 191.
- 6) 玉野・柳本：機論、**36**(1970), 126.

- 7) 柳本・佐々木・木内・河野：塑性と加工、**33**-383(1992), 1406.
- 8) Yanagimoto, J., Karhausen, K., Brand, A.J. and Kopp, R.: *Trans. ASME, J. Manufact. Sci. and Eng.*, **120**-2(1998), 316.
- 9) Yanagimoto, J. and Liu, J.: *ISIJ International*, **39**-2(1999), 171.
- 10) Yanagimoto, J., Ito, T. and Liu, J.: *ISIJ International*, **40**-1(2000), 65.
- 11) 玉野・柳本：機論、**41**-344(1975), 1130.
- 12) 柳本・木内・王・中野・川井：平 1 1 春塑加講論(1999), 355.
- 13) 柳本・柳本・青木：塑性と加工、**34**-395(1993), 1314.
- 14) 柳本：機論、**27**-178(1961), 800.
- 15) 戸澤・石川・岩田：塑性と加工、**23**-263(1982), 1181.
- 16) たとえば長田・柳本：基礎からわかる塑性加工(1997), コロナ社.
- 17) 美坂・吉本：塑性と加工、**8**(1967), 414.
- 18) Shohet, K.N. and Townsend, N.A.: *J. Iron and Steel Inst.*, **206**-11(1968), 1088.
- 19) 戸澤・上田：塑性と加工、**11**-108(1970), 29.
- 20) 中島・松本：24 回塑加連講論(1973), 29.
- 21) 第 92・93 回西山記念講座テキスト(1983), 日本鉄鋼協会.
- 22) 板圧延の理論と実際(1984), 日本鉄鋼協会.
- 23) 戸澤康壽：第 74 回塑性加工学講座テキスト, (1999), 13.
- 24) 中島・菊間・松本：材料, **30**-322 (1981), 524.